1 Direct Dynamics
Newton-Euler Equations of Motion

Problem 2

A slender rod $AB = L$ (link 2) is moving without friction along the slider 1(Fig. 1.5). The slider is connnected to the ground by a pin joint at O and is free to swing in a vertical plane. The mass of the rod is m_1 the mass center is located at C. The mass of the slider is m_2 and the mass moment of inertia of the slider with respect to its mass center point O is $I_O \approx 0$. The acceleration due to gravity is g. Find the Newton Euler equations of motion.

Numerical application: $m_1 = m_2 = 1$ kg, $L = 1$ m, $g = 10$ m/s^2.



Figure 5



```
(* Problem 2.a *)
(* TR *)
Apply [Clear,Names["Global`*"]];
Off[General::spell];
Off[General::spell1];

"Polar coordinates: n, t "

rC= {r[t], 0, 0} ;
Print["rC = ",rC];

F12= {0, F12t, 0} ;
Print["F12 = ", F12];
Print["The joint force F12 acts at the point P"];
Print["For link 1: IO alpha = OP x (-F12)"];
Print["if IO = 0 => OP=0, O=P => F12 acts at O"];

G2= {-m g Sin[theta[t]], -m g Cos[theta[t]], 0};
Print["G2 = ", G2];

alpha= {0, 0, theta''[t]};

aCr=r''[t] - r[t] (theta'[t])^2;
aCt=r[t]theta''[t] + 2 r'[t] theta'[t];
aC={aCr,aCt,0};
Print["aC = ", aC];

"m2 aC  = - F12 + G2 , (I) => "
e2= m aC  +  F12 - G2;
"(I)(on n) => Equation (1) is "
e2x=e2[[1]]==0
"(I)(on t) => Equation (2) is "
e2y=e2[[2]]==0

IC=m L^2/12;
Print["IC = ", IC];
"IC alpha = rCO x F12 , (3)"
e2=IC alpha-Cross[-rC,F12];
e2z=Simplify[e2[[3]]]==0

"From Eq.(3) => "
solf=Solve[e2z,F12t];
f12s=F12t/.solf[[1]];
Print["F12t = ",f12s];

"Substituting F12t in Eqs. (1) and (2) => two ODE "
eq1=e2x
eq2=e2y/.solf[[1]]


Print["numerical application"]

rule={m→1.,L→1,g→10.}

equation1=Simplify[eq1/.rule]
equation2=Simplify[eq2/.rule]

tf=.5;
Print["time interval = 0; ",tf];
Print["initial conditions: "]
Print["r[0]=.15, theta[0]=5 N[Pi]/180"];
Print["r'[0]=0., theta'[0]=0."];
sol=NDSolve[{equation1,equation2,r[0]==.15,theta[0]==5 N[Pi]/180,
        r'[0]==0.,theta'[0]==0.},{r,theta},{t,0.,tf}];
```

```
Plot[Evaluate[r[t]]/.sol,{t,0.,tf},PlotRange→All,
    AxesLabel→{"t[s]","r[m]"}];
Plot[Evaluate[theta[t]]/.sol,{t,0.,tf},PlotRange→All,
  AxesLabel→{"t[s]","theta[rad]"}]

 r5= (Evaluate[r[t]]/.sol/.t->.5)[[1]];
 t5= (Evaluate[theta[t]]/.sol/.t->.5)[[1]];
Print["r[.5] = ",r5, " m"];
Print["theta[.5] = ",t5, " rad"];
```

Polar coordinates: n, t

rC = {r[t], 0, 0}

F12 = {0, F12t, 0}

The joint force F12 acts at the point P

For link 1: IO alpha = OP x (-F12)

if IO = 0 => OP=0, O=P => F12 acts at O

G2 = {-g m Sin[theta[t]], -g m Cos[theta[t]], 0}

aC = $\{-r[t]\,theta'[t]^2 + r''[t],\ 2\,r'[t]\,theta'[t] + r[t]\,theta''[t],\ 0\}$

m2 aC = - F12 + G2 , (I) =>

(I)(on n) => Equation (1) is

$$g\,m\,Sin[theta[t]] + m\,(-r[t]\,theta'[t]^2 + r''[t]) == 0$$

(I)(on t) => Equation (2) is

$$F12t + g\,m\,Cos[theta[t]] + m\,(2\,r'[t]\,theta'[t] + r[t]\,theta''[t]) == 0$$

$$IC = \frac{L^2\,m}{12}$$

IC alpha = rCO x F12 , (3)

$$F12t\,r[t] + \frac{1}{12}\,L^2\,m\,theta''[t] == 0$$

From Eq.(3) =>

$$F12t = -\frac{L^2\,m\,theta''[t]}{12\,r[t]}$$

Substituting F12t in Eqs. (1) and (2) => two ODE

$$g\,m\,Sin[theta[t]] + m\,(-r[t]\,theta'[t]^2 + r''[t]) == 0$$

$$g\,m\,Cos[theta[t]] - \frac{L^2\,m\,theta''[t]}{12\,r[t]} + m\,(2\,r'[t]\,theta'[t] + r[t]\,theta''[t]) == 0$$

numerical application

$\{m \to 1.,\ L \to 1,\ g \to 10.\}$

$$1.\,Sin[theta[t]] + 0.1\,r''[t] == 0.1\,r[t]\,theta'[t]^2$$

1. $\text{Cos}[\text{theta}[t]] + 0.2\, r'[t]\, \text{theta}'[t] + 0.1\, r[t]\, \text{theta}''[t] == \frac{0.00833333\, \text{theta}''[t]}{r[t]}$

time interval = 0; 0.5

initial conditions:

r[0]=.15, theta[0]=5 N[Pi]/180

r'[0]=0., theta'[0]=0.





- Graphics -

r[.5] = -0.28339 m

theta[.5] = 2.34185 rad